

9/11.3

Le publipostage ou « mailing »

Le nom barbare de « mailing » désigne des lettres personnalisées écrites pour plusieurs destinataires différents.

La lettre de base (souvent appelée « modèle ») comporte des éléments de texte et des caractères spéciaux qui seront remplacés par des données issues d'un fichier de données.

Le programme proposé dans ce chapitre peut être utilisé au choix en Basic ou en Turbo Pascal. Les caractères de remplacement (caractères spéciaux) sont toujours "&&".

COMMENT UTILISER LE PROGRAMME

Si vous désirez travailler en Basic, saisissez le listing suivant :

```

1000 REM =====
1010 REM Mailings
1020 REM =====
1030 REM
1040 '
1050 ' - - - - -
1060 ' Dimensionnement des tableaux du programme
1070 ' - - - - -
1080 '
1090 MODE 1
1100 DIM mo$(65) 'Nombre de lignes du modele
1110 DIM do$(50) 'Nombre de lignes de donnees
1120 DIM x$(65) 'Tableau intermediaire
1130 PRINT"                CPC Mailing"
1140 PRINT"                -----"
1150 PRINT
1160 '
    
```

```
1170 ' - - - - -
1180 ' Parametrage de l'impression
1190 ' - - - - -
1200 '
1210 INPUT "Nom du modele : ",m$
1220 INPUT "Nom des donnees : ",d$
1230 INPUT "Nombre de lignes par feuille : ",taille
1240 PRINT
1250 numero=1 'Numero de la page imprimee
1260 '
1270 ' - - - - -
1280 ' Lecture des fichiers (modele et donnees)
1290 ' - - - - -
1300 '
1310 OPENIN m$
1320 i=0
1330 i=i+1
1340 LINE INPUT #9,mo$(i)
1350 IF EOF=0 THEN 1330
1360 i1=i 'Nombre de lignes dans le modele
1370 CLOSEIN
1380
1390 OPENIN d$
1400 i=0
1410 i=i+1
1420 LINE INPUT #9,do$(i)
1430 IF EOF=0 THEN 1410
1440 CLOSEIN
1450 i2=i 'Nombre de donnees
1460 '
1470 ' - - - - -
1480 ' Impression
1490 ' - - - - -
1500 '
1510 j=1 'Pointeur de donnee
1520 LOCATE 1,16
1530 PRINT"Impression en cours ...";
1540 PRINT " lettre No ";numero
1550 numero=numero+1
1560 '
1570 ' - - - - -
1580 ' Impression d'une lettre
1590 ' - - - - -
1600 '
1610 FOR i=i TO i1
1620 p=INSTR(mo$(i),"&&")
```

```

1630 IF p=0 THEN x#=mo$(i):GOTO 1680
1640 IF p=1 THEN x#=do$(j)
1650 IF p<>1 THEN x#=LEFT$(mo$(i),p-1)+do$(j)
1660 IF (LEN(mo$(i))-1-p)<>0 THEN x#=x#+RIGHT$(mo$(i),LEN(
mo$(i))-1-p)
1670 j=j+1
1680 PRINT #8,x#
1690 NEXT i
1700 '
1710 ' -----
1720 ' Lignes blanches en fin de lettre
1730 ' -----
1740 '
1750 FOR i=1 TO taille-1
1760 PRINT#8
1770 NEXT i
1780 IF j<12 THEN 1520 'Boucle d'impression
1790 PRINT:PRINT "Impression terminee."
1800 END

```

Si vous désirez travailler en Turbo Pascal, saisissez le listing suivant :

```

Program Mail;
{-----}
{ Réalisation de mailings }
{-----}

VAR
TMO : Array[1..65] of String[80]; { Tableau Modele }
TDO : Array[1..50] of String[80]; { Tableau de Donnees }
X : String[80]; { Une des lignes a imprimer }
M : String[8]; { Nom du modele }
D : String[8]; { Nom des donnees }
Paille : Byte; { Nombre de lignes par feuille }
Numero : Byte; { Numero de la page imprimée }
I : Byte; { Index de boucles }
J : Byte; { Pointeur de donnees }
I1 : Byte; { Nombre de lignes dans le modele }
I2 : Byte; { Nombre de donnees }
F : Text; { Modele et donnees }
P : Byte; { Position d'un && dans la ligne }

Procedure Init;
{-----}
{ Initialisation du programme }
{-----}

```

```

begin
  ClrScr;
  Writeln('                                CPC Mailing');
  Writeln('                                -----');
  Writeln;
  Write('Nom du modele : ');
  Readln(M);
  Write('Nom des donnees : ');
  Readln(D);
  Write('Nombre de lignes par feuille : ');
  Readln(Taille);
  Writeln;
  Numero:=1; { Numero de la page imprimee }
end;

```

```

Procedure Lit_Data;
{-----}
{ Lecture des fichiers : Modele et Donnees }
{-----}

```

```

begin
  Assign(F,M);
  Reset(F);
  I:=1;
  Repeat
    Readln(F,TMO[I]);
    I:=I+1;
  Until Eof(F);
  I1:=I-1; { Nombre de lignes dans le modele }
  Close(F);

  Assign(F,D);
  Reset(F);
  I:=1;
  Repeat
    Readln(F,TDO[I]);
    I:=I+1;
  Until Eof(F);
  I2:=I-1;
  Close(F); { Nombre de donnees }
end;

```

```

Procedure Imprime;

```

```

{-----}
{ Impression du modele en incorporant }
{ les donnees du fichier de donnees }
{-----}

begin
  J:=1;
  Repeat
    GotoXY(1,16);
    Write('Impression en cours ...');
    Write(' lettre No ',Numero);
    Numero:=Numero+1;
    { Impression d'une lettre }
    For I:=1 to I1 do
      begin
        P:=Pos('&&',TMO[I]);
        If P=0
          then
            X:=TMO[I]
          else
            begin
              If (P=1) then
                X:=TDO[J];
              If (P<>1)
                then X:=Concat(Copy(TMO[I],1,P-1),TDO[J]);
              If ((Length(TMO[I])-1-P)<>0) then
                X:=Concat(X,Copy(TMO[I],P+2,Length(TMO[I])-1-P));
              J:=J+1;
            end;
        WriteLn(Lst,X);
      end;
    {-----}
    { Lignes blanches en fin de lettre }
    {-----}
    For I:=1 to Taille-I1 do
      WriteLn(Lst);
  Until (J>I2);
  WriteLn;
  WriteLn('Impression terminee');
end;

```

```

{-----}
{ PROGRAMME PRINCIPAL }
{-----}

```

```

begin
  Init;      { Initialisation du programme }
  Lit_Data; { Lecture du modele et des donnees }
  Imprime;  { Impression du mailing }
end.

```

Avant de lancer le programme, vous devez créer un modèle et un fichier de données. Ces deux fichiers doivent contenir du texte au format ASCII. Vous les définirez avec un traitement de texte ou un éditeur de texte quelconque (Wordstar, éditeur du Turbo Pascal, etc.).

Exemple :

Le modèle pris en exemple représente une lettre adressée par une société à un client. Elle informe ce client que l'article commandé n'est pas disponible. Cinq caractères spéciaux permettent d'insérer dans chaque lettre :

- le nom et l'adresse du client ;
- la date courante ;
- le texte « Cher Monsieur, », si le client est un homme ou le texte « Chère Madame, » si le client est une femme.

Voici le contenu du modèle :

&&
&&
&&

Paris, le &&

&&,

Nous avons bien reçu votre courrier concernant votre commande. Malheureusement, l'article que vous avez choisi est momentanément indisponible. Soyez certains que nous faisons notre possible pour vous satisfaire au plus vite. Le délai d'attente est d'environ 2 semaines.

Nous restons à votre entière disposition pour tout renseignement complémentaire. Recevez nos salutations distinguées.

Michel Dubois, Responsable de la communication

Le fichier de données doit être composé d'un ou de plusieurs ensembles de cinq données. Chaque ensemble donnera lieu à une lettre. Dans notre exemple, le mailing sera adressé à trois personnes (15 données) :

```
Mr Jacques Dupuis  
12 rue Charles Huchet  
75008 Paris  
15 Juillet 1991  
Cher Monsieur  
Mme Gabrielle Reynes  
14 Residence du Bois  
91420 Morangis  
15 Juillet 1991  
Chere Madame  
Mr Gerard Paulin  
12 rue Guerin  
75012 Paris  
15 Juillet 1991  
Cher Monsieur
```

Voici le résultat du mailing :

Mr Jacques Dupuis
12 rue Charles Huchet
75008 Paris

Paris, le 15 Juillet 1991

Cher Monsieur,

Nous avons bien reçu votre courrier concernant votre commande. Malheureusement, l'article que vous avez choisi est momentanément indisponible. Soyez certains que nous faisons notre possible pour vous satisfaire au plus vite. Le délai d'attente est d'environ 2 semaines.

Nous restons à votre entière disposition pour tout renseignement complémentaire. Recevez nos salutations distinguées.

Mme Gabrielle Reynes
14 Residence du Bois
91420 Morangis

Paris, le 15 Juillet 1991

Chere Madame,

Nous avons bien reçu votre courrier concernant votre commande. Malheureusement, l'article que vous avez choisi est momentanément indisponible. Soyez certains que nous faisons notre possible pour vous satisfaire au plus vite. Le délai d'attente est d'environ 2 semaines.

Nous restons à votre entière disposition pour tout renseignement complémentaire. Recevez nos salutations distinguées.

Mr Gerard Paulin
12 rue Guerin
75012 Paris

Paris, le 15 Juillet 1991

Cher Monsieur,


Nous avons bien reçu votre courrier concernant votre commande. Malheureusement, l'article que vous avez choisi est momentanément indisponible. Soyez certains que nous faisons notre possible pour vous satisfaire au plus vite. Le délai d'attente est d'environ 2 semaines.

Nous restons à votre entière disposition pour tout renseignement complémentaire. Recevez nos salutations distinguées.

Le nom et l'adresse de chaque client est bien inséré à la place des trois premiers caractères spéciaux. La date apparaît en clair à la place du quatrième champ spécial. Enfin, la lettre commence par « Cher Monsieur, » si le client est un homme, et par « Chère Madame, » si le client est une femme.

LE PROGRAMME EN DÉTAIL

La logique de fonctionnement du programme apparaît dans l'ordino-gramme suivant :



Nous allons par exemple décrire le fonctionnement du programme écrit en Turbo Pascal. Vous n'aurez sans doute aucun mal à transférer les explications données au programme écrit en Basic si cela est nécessaire.

Le programme débute par la déclaration des variables.

Le tableau **TMO** est utilisé pour stocker les lignes du modèle. Dans le listing, ce tableau peut contenir 65 lignes de 80 caractères maximum. Modifiez ces valeurs si nécessaire :

```
TMO : Array[1..65] of String[80];
```

Le tableau **TDO** est utilisé pour stocker les données à appliquer au modèle. Dans le listing, ce tableau peut contenir 50 lignes de 80 caractères maximum. Modifiez ces valeurs si nécessaire :

```
TDO : Array[1..50] of String[80];
```

La variable **X** est utilisée par le programme pour convertir les caractères de contrôle. Sa longueur dans le programme est de 80 caractères. Modifiez cette valeur si les lignes imprimées peuvent avoir plus de 80 caractères :

```
X : String[80];
```

Les variables **M** et **D** représentent respectivement le nom du modèle et le nom du fichier de données. Leur taille maximale est de 8 caractères. Modifiez ces valeurs si nécessaire :

```
M : String[8];
```

```
D : String[8];
```

Le programme se poursuit par la section des procédures.

La procédure **Init** acquiert :

– le nom du modèle :

```
Write ('Nom du modèle : ');  
Readln(M);
```

– le nom du fichier de données :

```
Write ('Nom des données : ');  
Readln(D);
```

– le nombre de lignes par feuille. Il s'agit du nombre de lignes imprimables (par exemple 66 sur une page 21x28 cm), et non du nombre de lignes du modèle. Ce paramètre est utilisé par le programme pour calculer le nombre de lignes blanches à sauter entre deux lettres consécutives.

```
Write ('Nombre de lignes par feuille : ');
Readln(Taille);
```

La variable "Numero" contient le numéro de la page imprimée. Elle est initialisée à 1 dans la procédure Init.

Comme son nom l'indique, la procédure Lit_Data lit le contenu des fichiers modèle et données et les stocke en mémoire.

Les données sont lues à l'intérieur d'une boucle Repeat Until qui prend fin lorsque la fin du fichier (caractère Eof) a été atteinte :

```
I:=1;
Repeat
  Readln(F,TMO[I]);
  I:=I+1;
Until Eof(F);
```

Le nombre de lignes dans le modèle et le nombre de lignes de données sont mémorisés dans les variables I1 et I2 :

```
I1:=I-1;
...
I2:=I-1;
```

La procédure Imprime incorpore les données du fichier de données dans le modèle et imprime les lettres type.

Les I1 lignes du modèle sont analysées dans le but d'identifier un caractère spécial "&&" :

```
For I:=1 to I1 do
begin
  P:=Pos('&&',TMO[I]);
```

Lorsqu'aucun caractère spécial n'est rencontré dans la ligne, elle est stockée dans la variable X :

```
If P=0
then
  X:=TMO[I];
```

Lorsqu'un caractère spécial est rencontré dans la ligne, il est remplacé par la donnée correspondante.

Le traitement dépend de la position du caractère spécial :

– s'il se trouve en début de ligne, la donnée est stockée dans la variable X :

```
If (P=1) then
  x:=TDO[J];
```

– dans le cas contraire, la donnée est précédée d'une portion de texte :

```
If (P<>1)
  then x:=Concat (Copy(TMO[I],1,P-1), TDO[J]);
```

– lorsque le caractère spécial est suivi d'un texte, il faut le concaténer dans la variable x :

```
If ((Length(TMO[I])-1-P)<>0) then
  x:=Concat(x,Copy (TMO[I],P+2,Length(TMO[I])-1-P);
```

La position dans le fichier de données est incrémentée :

```
J:=J+1;
```

et la ligne ainsi modifiée est imprimée :

```
WriteLn(Lst,X);
```

Lorsque les I1 lignes ont été imprimées, la procédure Imprime complète l'impression par Taille-I1 lignes blanches :

```
For I:=1 to Taille-I1 do
  WriteLn(Lst);
```

La boucle d'impression prend fin lorsque toutes les données du fichier de données ont été utilisées :

```
...
Until (J>I2);
```

Le programme se termine par la section du programme principal qui active séquentiellement les trois procédures principales du programme :

```
Init;      { Initialisation du programme      }
Lit_Data;  { Lecture du modèle et des données }
Imprime;   { Impression du mailing           }
```